January 25, 2015 (unfinished); Comments are welcome!
junyongkim@snu.ac.kr, http://www.junyongkim.com/

Petersen, Mitchell A., 2009, "Estimating Standard Errors in Finance Panel Data Sets: Comparing Approaches," *Review of Financial Studies*, vol. 22, no. 1, pp. 436–480

Table 1. (Replication, 1,000 simulated panel data with 5,000 observations)

$Avg(\beta_{OLS})$ $Std(\beta_{OLS})$ $Avg(SE_{OLS})$ % $Sig(T_{OLS})$ $Avg(SE_C)$ % $Sig(T_C)$		Source of independent variable volatility			
		0%	25%	50%	75%
Source of residual volatility	0%	0.9988 0.0281 0.0283 [0.0090] 0.0283 [0.0070]	0.9999 0.0293 0.0283 [0.0150] 0.0283 [0.0140]	1.0005 0.0296 0.0283 [0.0150] 0.0283 [0.0140]	1.0011 0.0296 0.0284 [0.0100] 0.0283 [0.0100]
	25%	0.9984 0.0281 0.0283 [0.0080] 0.0283 [0.0070]	1.0006 0.0353 0.0283 [0.0430] 0.0353 [0.0120]	1.0017 0.0405 0.0283 [0.0760] 0.0412 [0.0120]	1.0027 0.0449 0.0283 [0.1090] 0.0463 [0.0100]
	50%	0.9984 0.0283 0.0283 [0.0120] 0.0283 [0.0130]	1.0009 0.0405 0.0283 [0.0710] 0.0411 [0.0070]	1.0021 0.0494 0.0283 [0.1420] 0.0508 [0.0070]	1.0032 0.0567 0.0283 [0.2000] 0.0591 [0.0080]
	75%	0.9985 0.0285 0.0283 [0.0130] 0.0283 [0.0140]	1.0011 0.0452 0.0283 [0.1200] 0.0462 [0.0070]	1.0024 0.0570 0.0283 [0.2040] 0.0589 [0.0080]	1.0036 0.0666 0.0283 [0.2880] 0.0695 [0.0070]

Table 2. (Replication, 1,000 simulated panel data with 5,000 observations)

$Avg(\beta_{FM})$ $Std(\beta_{FM})$ $Avg(SE_{FM})$ % $Sig(T_{FM})$		Source of independent variable volatility			
		0%	25%	50%	75%
Source of residual volatility	0%	0.9988 0.0283 0.0276 [0.0290]	1.0000 0.0295 0.0278 [0.0310]	1.0006 0.0298 0.0279 [0.0280]	1.0011 0.0297 0.0279 [0.0300]
	25%	0.9984 0.0281 0.0276 [0.0230]	1.0007 0.0355 0.0269 [0.0620]	1.0018 0.0406 0.0261 [0.1220]	1.0027 0.0449 0.0252 [0.1800]
	50%	0.9984 0.0283 0.0276 [0.0290]	1.0009 0.0406 0.0259 [0.1200]	1.0022 0.0494 0.0241 [0.2220]	1.0033 0.0568 0.0220 [0.3120]
	75%	0.9985 0.0286 0.0275 [0.0370]	1.0012 0.0452 0.0249 [0.1800]	1.0025 0.0570 0.0219 [0.3220]	1.0036 0.0667 0.0184 [0.4760]

Figure 2. (Replication, 1,000 simulated panel data with 5,000 observations)

Figure 2



Figure 2



Figure 2



Figure 3. (Replication, 1,000 simulated panel data with 5,000 observations)



Figure 4. (Replication, 1,000 simulated panel data with 5,000 observations)



(NOTE: IMPERFECT! SAS does not provide a panel Newey-West Standard Error!)

Table 3. (Replication, 1,000 simulated panel data with 5,000 observations)

$Avg(\beta_{OLS})$ $Std(\beta_{OLS})$ $Avg(SE_{OLS})$ % $Sig(T_{OLS})$ $Avg(SE_C)$ % $Sig(T_C)$		Source of independent variable volatility			
		0%	25%	50%	75%
Source of residual volatility	0%	0.9983 0.0281 0.0283 [0.0080] 0.0274 [0.0220]	0.9993 0.0277 0.0283 [0.0120] 0.0272 [0.0320]	0.9993 0.0281 0.0286 [0.0100] 0.0270 [0.0360]	0.9992 0.0290 0.0292 [0.0130] 0.0266 [0.0440]
	25%	0.9998 0.0272 0.0282 [0.0070] 0.0275 [0.0210]	0.9950 0.1637 0.0282 [0.6510] 0.1421 [0.0360]	0.9938 0.2298 0.0284 [0.7540] 0.1942 [0.0530]	0.9932 0.2898 0.0288 [0.8040] 0.2388 [0.0720]
	50%	1.0001 0.0270 0.0280 [0.0070] 0.0274 [0.0190]	0.9933 0.2309 0.0279 [0.7550] 0.1989 [0.0350]	0.9916 0.3246 0.0280 [0.8270] 0.2731 [0.0560]	0.9908 0.4092 0.0284 [0.8610] 0.3365 [0.0730]
	75%	1.0004 0.0270 0.0278 [0.0080] 0.0270 [0.0200]	0.9920 0.2828 0.0275 [0.8050] 0.2427 [0.0350]	0.9900 0.3975 0.0275 [0.8630] 0.3339 [0.0570]	0.9891 0.5011 0.0278 [0.8860] 0.4116 [0.0740]

Table 4. (Replication, 1,000 simulated panel data with 5,000 observations)

$Avg(\beta_{FM})$ $Std(\beta_{FM})$ $Avg(SE_{FM})$ % $Sig(T_{FM})$		Source of independent variable volatility			
		0%	25%	50%	75%
Source of residual volatility	0%	0.9993 0.0281 0.0275 [0.0240]	0.9993 0.0284 0.0284 [0.0300]	0.9991 0.0307 0.0311 [0.0260]	0.9989 0.0366 0.0375 [0.0200]
	25%	0.9999 0.0272 0.0276 [0.0170]	0.9966 0.1354 0.1241 [0.0270]	0.9973 0.1855 0.1712 [0.0320]	1.0001 0.2566 0.2394 [0.0340]
	50%	1.0002 0.0270 0.0274 [0.0170]	0.9956 0.1905 0.1729 [0.0250]	0.9966 0.2616 0.2399 [0.0300]	1.0007 0.3620 0.3362 [0.0350]
	75%	1.0005 0.0271 0.0271 [0.0200]	0.9949 0.2333 0.2108 [0.0260]	0.9962 0.3205 0.2929 [0.0330]	1.0013 0.4433 0.4107 [0.0350]

Figure. (Does not exist in the original paper—just for comparison)



OLS is miserable; Fama-MacBeth is noticeable!

SAS Code

```
resetline;
ods html close;
ods graphics off;
ods listing;

%let NSAMPLE=1000;
%let SIGMAX=1;
%let SIGMAEPS=2;

%macro FIRMEFFECT(INDS=,TIMES=,START=,FINISH=,INTERVAL=);

%do SOURCEX=&START. %to &FINISH. %by &INTERVAL.;

%do SOURCEEPS=&START. %to &FINISH. %by &INTERVAL.;

proc printto;
run;

%put INDS=&INDS. TIMES=&TIMES. SOURCEX=0.&SOURCEX. SOURCEε=0.&SOURCEEPS.;

proc printto log="nul:";
run;

data _01;
  do SAMPLE=1 to &NSAMPLE.;
  do IND=1 to &INDS.;
    MU=&SIGMAX.*sqrt(0.&SOURCEX.)*rannor(1);
    GAMMA=&SIGMAEPS.*sqrt(0.&SOURCEEPS.)*rannor(2);
    do TIME=1 to &TIMES.;
      X=MU+&SIGMAX.*sqrt(1-0.&SOURCEX.)*rannor(3);
      EPS=GAMMA+&SIGMAEPS.*sqrt(1-0.&SOURCEEPS.)*rannor(4);
      Y=X+EPS;
      output;
    end;
  end;
  end;
run;

proc reg noprint outest=_02 tableout;
  model Y=X/noint;
  by SAMPLE;
run;

proc transpose out=_02(rename=(COL1=BETAOLS COL2=SEOLS));
  var X;
  by SAMPLE;
  where _TYPE_ in ("PARMS","STDERR");
```

```
run;

ods listing close;
ods results off;
ods output parameterestimates=_03;

proc surveyreg data=_01;
  model Y=X/noint;
  cluster IND;
  by SAMPLE;
run;

ods listing;
ods results on;

proc sort data=_01;
  by SAMPLE TIME IND;
run;

proc reg noprint outest=_04;
  model Y=X/noint;
  by SAMPLE TIME;
run;

proc means noprint;
  var X;
  by SAMPLE;
  output out=_05 mean=BETAFM stderr=SEFM;
run;

data _02;
  merge _02 _03(where=(Parameter="X")) _05;
  by SAMPLE;
  TOLS=(BETAOLS-1)/SEOLS;
  SIGTOLS=(abs(TOLS)>2.58);
  SEC=StdErr;
  TC=(BETAOLS-1)/SEC;
  SIGTC=(abs(TC)>2.58);
  TFM=(BETAFM-1)/SEFM;
  SIGTFM=(abs(TFM)>2.58);
  keep SAMPLE BETAOLS SEOLS TOLS SIGTOLS SEC TC SIGTC BETAFM SEFM TFM SIGTFM;
run;

proc means data=_02 noprint;
  var BETAOLS SEOLS SIGTOLS SEC SIGTC BETAFM SEFM SIGTFM;
  output out=_06
    mean=BETAOLS SEOLS SIGTOLS SEC SIGTC BETAFM SEFM SIGTFM
    std=S_BETAOLS S_SEOLS S_SIGTOLS S_SEC S_SIGTC S_BETAFM S_SEFM S_SIGTFM;
run;
```

```
%if &SOURCEX.=50 and &SOURCEEPS.=50 and &INTERVAL.=25 %then %do;

goptions xpixels=800 ypixels=400 border;
title "Figure 2";

proc univariate data=_02;
  var TOLS TC TFM;
  histogram/normal(mu=0 sigma=1) midpoints=-7 to 7 vaxislabel="Percentile";
run;

goptions;
title "The SAS System";

%end;

data _06;
  retain BETAOLS S_BETAOLS BETAFM S_BETAFM;
  set _06;
  UNDEROLS=1-SEOLS/S_BETAOLS;
  UNDERC=1-SEC/S_BETAOLS;
  UNDERFM=1-SEFM/S_BETAFM;
  INDS=&INDS.;
  TIMES=&TIMES.;
  SOURCEX=0.&SOURCEX.;
  SOURCEEPS=0.&SOURCEEPS.;
run;

proc append base=_07 data=_06;
run;

%end;

%end;

%mend;

/**/

proc printto log="nul:";
run;

%FIRMEFFECT(INDS=500,TIMES=10,START=0,FINISH=75,INTERVAL=25);

proc printto;
run;

proc transpose out=_03;
  var BETAOLS--UNDERFM;
```

```
  by INDS TIMES SOURCEX SOURCEEPS;
run;

title "Table 1";

proc print noobs;
  var _NAME_ COL1;
  where TIMES=10 and _NAME_ in ("BETAOLS","S_BETAOLS","SEOLS","SIGTOLS","SEC","SIGTC");
  by SOURCEX SOURCEEPS;
  format COL1 8.4;
run;

title "Table 2";

proc print noobs;
  var _NAME_ COL1;
  where TIMES=10 and _NAME_ in ("BETAFM","S_BETAFM","SEFM","SIGTFM");
  by SOURCEX SOURCEEPS;
  format COL1 8.4;
run;

title "The SAS System";

proc sql;
  create table _A01_02 as select * from _02;
  create table _A01_03 as select * from _03;
  drop table _07;
quit;

/**/

proc printto log="nul:";
run;

%FIRMEFFECT(INDS=1000,TIMES=5,START=50,FINISH=50,INTERVAL=1);
%FIRMEFFECT(INDS=500,TIMES=10,START=50,FINISH=50,INTERVAL=1);
%FIRMEFFECT(INDS=250,TIMES=20,START=50,FINISH=50,INTERVAL=1);
%FIRMEFFECT(INDS=125,TIMES=40,START=50,FINISH=50,INTERVAL=1);
%FIRMEFFECT(INDS=100,TIMES=50,START=50,FINISH=50,INTERVAL=1);

proc printto;
run;

goptions xpixels=800 ypixels=600 border;
title "Figure 3";
symbol i=join v=circle;
legend position=(bottom center outside) frame;

proc gplot;
```

```
  plot UNDEROLS*TIMES UNDERC*TIMES UNDERFM*TIMES/overlay legend=legend1;
run;

goptions;
title "The SAS System";
symbol;
legend;

proc sql;
  create table _A02_02 as select * from _02;
  create table _A02_07 as select * from _07;
  drop table _01,_02,_03,_04,_05,_06,_07;
quit;

/**/

proc printto log="nul:";
run;

%FIRMEFFECT(INDS=500,TIMES=10,START=25,FINISH=25,INTERVAL=1);

proc printto;
run;

proc sort data=_01;
  by SAMPLE IND TIME;
run;

%macro NEWEYWEST;

%do NWLAG=0 %to 9;

proc printto;
run;

%put NWLAG=&NWLAG.;

proc printto log="nul:";
run;

ods listing close;
ods results off;
ods output parameterestimates=_08;

proc model data=_01;
  Y=BETA0+BETA1*X;
  parameters BETA0 BETA1;
  fit Y/gmm kernel=(bart,%eval(&NWLAG.+1),0) vardef=n;
  by SAMPLE;
```

```
run;

ods listing;
ods results on;

proc sql;
  create table _08 as
  select &NWLAG. as NWLAG,SAMPLE,StdErr as SENW label=""
  from _08(where=(Parameter="BETA1"));
quit;

proc append base=_09 data=_08;
run;

%end;

%mend;

proc printto log="nul:";
run;

%NEWEYWEST;

proc printto;
run;

proc means noprint;
  var SENW;
  by NWLAG;
  output out=_10 mean=SENW;
run;

proc sql;
  create table _03 as
  select NWLAG,S_BETAOLS+1e-4 as S_BETAOLS,SEOLS,SEC-1e-4 as SEC,SENW
  from _06,_10
  where _06._TYPE_=_10._TYPE_;
quit;

goptions xpixels=800 ypixels=600 border;
title "Figure 4";
symbol i=join v=circle;
legend position=(bottom center outside) frame;

proc gplot;
  plot S_BETAOLS*NWLAG SEOLS*NWLAG SEC*NWLAG SENW*NWLAG/overlay legend=legend1;
run;

goptions;
```

```
title "The SAS System";
symbol;
legend;

proc sql;
  create table _A03_03 as select * from _03;
  create table _A03_09 as select * from _09;
  drop table _01,_02,_03,_04,_05,_06,_07,_08,_09,_10;
quit;

/**/

%macro TIMEEFFECT(INDS=,TIMES=,START=,FINISH=,INTERVAL=);

%do SOURCEX=&START. %to &FINISH. %by &INTERVAL.;

%do SOURCEEPS=&START. %to &FINISH. %by &INTERVAL.;

proc printto;
run;

%put INDS=&INDS. TIMES=&TIMES. SOURCEX=0.&SOURCEX. SOURCEε=0.&SOURCEEPS.;

proc printto log="nul:";
run;

data _01;
  do SAMPLE=1 to &NSAMPLE.;
  do TIME=1 to &TIMES.;
    ZETA=&SIGMAX.*sqrt(0.&SOURCEX.)*rannor(1);
    DELTA=&SIGMAEPS.*sqrt(0.&SOURCEEPS.)*rannor(2);
    do IND=1 to &INDS.;
      X=ZETA+&SIGMAX.*sqrt(1-0.&SOURCEX.)*rannor(3);
      EPS=DELTA+&SIGMAEPS.*sqrt(1-0.&SOURCEEPS.)*rannor(4);
      Y=X+EPS;
      output;
    end;
  end;
  end;
run;

proc reg noprint outest=_02 tableout;
  model Y=X/noint;
  by SAMPLE;
run;

proc transpose out=_02(rename=(COL1=BETAOLS COL2=SEOLS));
  var X;
  by SAMPLE;
```

```
  where _TYPE_ in ("PARMS","STDERR");
run;

ods listing close;
ods results off;
ods output parameterestimates=_03;

proc surveyreg data=_01;
  model Y=X/noint;
  cluster TIME;
  by SAMPLE;
run;

ods listing;
ods results on;

proc reg noprint data=_01 outest=_04;
  model Y=X/noint;
  by SAMPLE TIME;
run;

proc means noprint;
  var X;
  by SAMPLE;
  output out=_05 mean=BETAFM stderr=SEFM;
run;

data _02;
  merge _02 _03(where=(Parameter="X")) _05;
  by SAMPLE;
  TOLS=(BETAOLS-1)/SEOLS;
  SIGTOLS=(abs(TOLS)>2.58);
  SEC=StdErr;
  TC=(BETAOLS-1)/SEC;
  SIGTC=(abs(TC)>2.58);
  TFM=(BETAFM-1)/SEFM;
  SIGTFM=(abs(TFM)>2.58);
  keep SAMPLE BETAOLS SEOLS TOLS SIGTOLS SEC TC SIGTC BETAFM SEFM TFM SIGTFM;
run;

proc means data=_02 noprint;
  var BETAOLS SEOLS SIGTOLS SEC SIGTC BETAFM SEFM SIGTFM;
  output out=_06
    mean=BETAOLS SEOLS SIGTOLS SEC SIGTC BETAFM SEFM SIGTFM
    std=S_BETAOLS S_SEOLS S_SIGTOLS S_SEC S_SIGTC S_BETAFM S_SEFM S_SIGTFM;
run;

%if &SOURCEX.=50 and &SOURCEEPS.=50 and &INTERVAL.=25 %then %do;
```

```
goptions xpixels=800 ypixels=400 border;
title "Figure 2";

proc univariate data=_02;
  var TOLS;
  histogram/normal(mu=0 sigma=1) midpoints=-50 to 50 by 5 vaxislabel="Percentile";
run;

proc univariate data=_02;
  var TC TFM;
  histogram/normal(mu=0 sigma=1) midpoints=-6 to 6 vaxislabel="Percentile";
run;

goptions;
title "The SAS System";

%end;

data _06;
  retain BETAOLS S_BETAOLS BETAFM S_BETAFM;
  set _06;
  UNDEROLS=1-SEOLS/S_BETAOLS;
  UNDERC=1-SEC/S_BETAOLS;
  UNDERFM=1-SEFM/S_BETAFM;
  INDS=&INDS.;
  TIMES=&TIMES.;
  SOURCEX=0.&SOURCEX.;
  SOURCEEPS=0.&SOURCEEPS.;
run;

proc append base=_07 data=_06;
run;

%end;

%end;

%mend;

/**/

proc printto log="nul:";
run;

%TIMEEFFECT(INDS=500,TIMES=10,START=0,FINISH=75,INTERVAL=25);

proc printto;
run;
```

```
proc transpose out=_03;
  var BETAOLS--UNDERFM;
  by INDS TIMES SOURCEX SOURCEEPS;
run;

title "Table 3";

proc print noobs;
  var _NAME_ COL1;
  where TIMES=10 and _NAME_ in ("BETAOLS","S_BETAOLS","SEOLS","SIGTOLS","SEC","SIGTC");
  by SOURCEX SOURCEEPS;
  format COL1 8.4;
run;

title "Table 4";

proc print noobs;
  var _NAME_ COL1;
  where TIMES=10 and _NAME_ in ("BETAFM","S_BETAFM","SEFM","SIGTFM");
  by SOURCEX SOURCEEPS;
  format COL1 8.4;
run;

title "The SAS System";

proc sql;
  create table _A04_02 as select * from _02;
  create table _A04_03 as select * from _03;
  drop table _07;
quit;
```